One-locus two-deme model of migration and selection with deme-independent dominance

Life cycle

Random mating → viability selection → population regulation → migration

Model and nomenclature

We consider a single locus with two alleles A_1 and A_2 and denote the frequencies of these two alleles in deme 1 and 2 by p_1 , $q_1 = 1 - p_1$ and p_2, $q_2 = 1 - p_2$, respectively. Each generation after population regulation and before random mating, a proportion m_{ij} of individuals in deme i is replaced by immigrants from deme j.

We denote the relative fitness of genotype $A_k A_l$ in deme i by w_{ikl} and parameterise these fitnesses as follows:

$w_{111} = 1$

$w_{112} = w_{121} = 1 - h\, s_1$

$w_{122} = 1 - s_1$

$w_{211} = 1 - s_2$

$w_{212} = w_{221} = 1 - h\, s_2$

$w_{222} = 1$

The fitness parameterisation above is fully symmetrical with respect to the two demes and assumes that dominance is independent of the deme. Because relative fitnesses are bound to be positive, selection coefficients are constrained to be $0 \le s_i \le 1$. Moreover, we assume that there is no underdominance nor overdominance, i.e. that $0 \le h \le 1$. As a consequence, if $s_i > 0$, then A_i confers a selective advantage in deme i and a disadvantage in deme $j \neq i$. Note that h here reflects the dominance of the locally maladaptive allele, i.e. of A_i in deme $j \neq i$.

Below, we also implement the parameterisation used by Sibly et al. (2021):

$w_{111} = 1$

$w_{112} = w_{121} = 1 + h\, s_1$

$w_{122} = 1 + s_1$

$w_{211} = 1$

$w_{212} = w_{221} = 1 + h\, s_2$

$w_{222} = 1 + s_2$

Assumptions

- Deterministic dynamics
- Infinite population sizes
- Deme-independent dominance
- No underdominance nor overdominance
- We assume no parental effect on the fitness of heterozygotes, i.e. $w_{kij} = w_{kji}$ for $i \neq j$.

Generic functions

In[56]:=
```
p[k_] := If[k == 1, p1, p2]
q[k_] := 1 - p[k]
```

In[58]:=
```
p[1]
```
Out[58]=
```
p1
```

In[59]:=
```
q[2]
```
Out[59]=
```
1 - p2
```

Fitness functions and rules

Relative fitness rules

In[60]:=
```
wMat := {
  {{w111, w112}, {w121, w122}},
  {{w211, w212}, {w221, w222}}
 }
```

In[61]:=
```
w[k_, i_, j_] := wMat[[k, i, j]]
```

In[62]:=
```
fitRuleSymmet := {
  w111 → 1, w112 → 1 - h s1, w121 → 1 - h s1, w122 → 1 - s1,
  w211 → 1 - s2, w212 → 1 - h s2, w221 → 1 - h s2, w222 → 1
 }

fitRuleSibly := {
  w111 → 1, w112 → 1 + h s1, w121 → 1 + h s1, w122 → 1 + s1,
  w211 → 1, w212 → 1 + h s2, w221 → 1 + h s2, w222 → 1 + s2
 }
```

In[64]:=
```
w[1, 2, 1] /. fitRuleSymmet
```
Out[64]= 1 - h s1

In[65]:=
```
w[1, 2, 1] /. fitRuleSibly
```
Out[65]= 1 + h s1

Marginal fitness function

In[66]:=
```
w[k_, i_] := If[i == 1,
   p[k] × w[k, 1, 1] + q[k] × w[k, 1, 2],
   p[k] × w[k, 1, 2] + q[k] × w[k, 2, 2]
  ]
```

In[67]:=
```
w[2, 1]
```
Out[67]= p2 w211 + (1 - p2) w212

Mean fitness function

In[68]:=
```
w[k_] := p[k] × w[k, 1] + q[k] × w[k, 2]
```

Generic rules

In[69]:=
```
qRule := {q1 → 1 - p1, q2 → 1 - p2}
```

Migration rules

In[70]:=
```
cRule := {c1 → 1 + m21 - m12, c2 → 1 + m12 - m21}
symmetMigRule := {m12 → m, m21 → m}
```

Dynamical equations

Recursion equations

Selection (Eq. 1)

In[72]:=

$$\text{p1Sel} := \text{p1}\,\frac{\text{w[1, 1]}}{\text{w[1]}}$$

$$\text{p2Sel} := \text{p2}\,\frac{\text{w[2, 1]}}{\text{w[2]}}$$

In[74]:=
```
p1SelSibly = p1Sel /. fitRuleSibly // Simplify
p2SelSibly = p2Sel /. fitRuleSibly // Simplify
```

Out[74]= $$\frac{\text{p1}\ (-1+\text{h}\ (-1+\text{p1})\ \text{s1})}{-1+(-1+\text{p1})\times(1+(-1+2\ \text{h})\ \text{p1})\ \text{s1}}$$

Out[75]= $$\frac{\text{p2}\ (-1+\text{h}\ (-1+\text{p2})\ \text{s2})}{-1+(-1+\text{p2})\times(1+(-1+2\ \text{h})\ \text{p2})\ \text{s2}}$$

In[76]:=
```
p1SelSymmet = p1Sel /. fitRuleSymmet // Simplify
p2SelSymmet = p2Sel /. fitRuleSymmet // Simplify
```

Out[76]= $$\frac{\text{p1}+\text{h}\ (-1+\text{p1})\ \text{p1}\ \text{s1}}{1+(-1+\text{p1})\times(1+(-1+2\ \text{h})\ \text{p1})\ \text{s1}}$$

Out[77]= $$\frac{\text{p2}\ (1+\text{h}\ (-1+\text{p2})\ \text{s2}-\text{p2}\ \text{s2})}{1+2\ \text{h}\ (-1+\text{p2})\ \text{p2}\ \text{s2}-\text{p2}^2\ \text{s2}}$$

As a check, we may use the symmetry of the model, noting that $p_1 = 1 - p_2$:

In[78]:=
```
p1SelSymmet - (1 - (p2SelSymmet /. {p2 → 1 - p2} /. {p2 → p1} /. {s2 → s1})) //
 FullSimplify
```

Out[78]= 0

In[79]:=

$$\text{p1SelSiblyGiven} := \frac{\text{p1}^2+(1+\text{h}\ \text{s1})\ \text{p1}\ \text{q1}}{1+2\ \text{h}\ \text{p1}\ \text{q1}\ \text{s1}+\text{s1}\ \text{q1}^2}$$

$$\text{p2SelSiblyGiven} := \frac{\text{p2}^2+(1+\text{h}\ \text{s2})\ \text{p2}\ \text{q2}}{1+2\ \text{h}\ \text{p2}\ \text{q2}\ \text{s2}+\text{s2}\ \text{q2}^2}$$

In[81]:=
```
p1SelSibly - (p1SelSiblyGiven /. q1 → (1 - p1)) // FullSimplify
p2SelSibly - (p2SelSiblyGiven /. q2 → (1 - p2)) // FullSimplify
```

Out[81]= 0

Out[82]= 0

Validation of Equations 2 to 5 of Sibly et al. (2021) Theor Popul Biol

In[83]:=

Equation 2:

In[84]:=

$$\text{p1PrimePrime} := \frac{(1-\text{m12})\ \text{p1Prime}+\text{m21}\ \text{p2Prime}}{\text{c1}}$$

$$\text{p2PrimePrime} := \frac{(1-\text{m21})\ \text{p2Prime}+\text{m12}\ \text{p1Prime}}{\text{c2}}$$

Assuming equilibrium, i.e. that $p_i'' = p_i$, and solving for p_1' yields Eq. (3).

In[86]:=
```
p1PrimeEq3LRule = Solve[p1PrimePrime == p1, p1Prime]
```

Out[86]= $\left\{\left\{\text{p1Prime} \to \frac{-\text{c1 p1} + \text{m21 p2Prime}}{-1 + \text{m12}}\right\}\right\}$

In[87]:=
```
p1PrimeEq3RRule = Solve[p2PrimePrime == p2, p1Prime]
```

Out[87]= $\left\{\left\{\text{p1Prime} \to \frac{\text{c2 p2} - \text{p2Prime} + \text{m21 p2Prime}}{\text{m12}}\right\}\right\}$

Equating the two expressions for p_1' and solving for p_2' yields Eq. (4a).

In[88]:=
```
K2Rule =
  Solve[(p1Prime /. p1PrimeEq3LRule) == (p1Prime /. p1PrimeEq3RRule), p2Prime]
```

Out[88]= $\left\{\left\{\text{p2Prime} \to \frac{\text{c1 m12 p1} - \text{c2 p2} + \text{c2 m12 p2}}{-1 + \text{m12} + \text{m21}}\right\}\right\}$

In[89]:=
```
p2PrimeEq3LRule = Solve[p2PrimePrime == p2, p2Prime]
```

Out[89]= $\left\{\left\{\text{p2Prime} \to \frac{\text{m12 p1Prime} - \text{c2 p2}}{-1 + \text{m21}}\right\}\right\}$

In[90]:=
```
p2PrimeEq3RRule = Solve[p1PrimePrime == p1, p2Prime]
```

Out[90]= $\left\{\left\{\text{p2Prime} \to \frac{\text{c1 p1} - \text{p1Prime} + \text{m12 p1Prime}}{\text{m21}}\right\}\right\}$

Equating the two expressions for p_2' and solving for p_1' yields Eq. (4b).

In[91]:=
```
K1Rule =
  Solve[(p2Prime /. p2PrimeEq3LRule) == (p2Prime /. p2PrimeEq3RRule), p1Prime]
```

Out[91]= $\left\{\left\{\text{p1Prime} \to \frac{-\text{c1 p1} + \text{c1 m21 p1} + \text{c2 m21 p2}}{-1 + \text{m12} + \text{m21}}\right\}\right\}$

Defining K_i according to Eq. (4):

In[92]:=
```
K1 = p1Prime /. K1Rule[[1]]
K2 = p2Prime /. K2Rule[[1]]
```

Out[92]= $\frac{-\text{c1 p1} + \text{c1 m21 p1} + \text{c2 m21 p2}}{-1 + \text{m12} + \text{m21}}$

Out[93]= $\frac{\text{c1 m12 p1} - \text{c2 p2} + \text{c2 m12 p2}}{-1 + \text{m12} + \text{m21}}$

Above, we expressed the allele frequencies after selection and before migration in terms of allele frequencies after migration, assuming equilibrium. In Eq. 1 we already expressed allele frequencies after selection in terms of allele frequencies before selection. Next, we equate the two expressions for p_i' and solve for the selection coefficients.

Using the parameterisation of Sibly et al. (2021)

In[94]:=
```
s1RuleSibly = Solve[κ1 == p1SelSiblyGiven, s1]
```

Out[94]= $\left\{\left\{\text{s1} \to \frac{\text{p1}^2 + \text{p1 q1} - \kappa\text{1}}{\text{q1}\ (-\text{h p1} + 2\ \text{h p1}\ \kappa\text{1} + \text{q1}\ \kappa\text{1})}\right\}\right\}$

In[95]:=
```
p1^2 + p1 q1 - κ1 /. q1 → (1 - p1) // FullSimplify
```
Out[95]= $p1 - \kappa 1$

In[96]:=
```
s2RuleSibly = Solve[κ2 == p2SelSiblyGiven, s2]
```
Out[96]= $\left\{\left\{s2 \to \frac{p2^2 + p2\, q2 - \kappa 2}{q2\,(-h\, p2 + 2\, h\, p2\, \kappa 2 + q2\, \kappa 2)}\right\}\right\}$

In[97]:=
```
p2^2 + p2 q2 - κ2 /. q2 → (1 - p2) // FullSimplify
```
Out[97]= $p2 - \kappa 2$

The same as above, but substituting $q_i \to (1 - p_i)$:

In[98]:=
```
s1RuleSiblyAlt = Solve[κ1 == p1SelSiblyGiven /. q1 → 1 - p1, s1]
```
Out[98]= $\left\{\left\{s1 \to \frac{-p1 + \kappa 1}{(-1 + p1)\,(-h\, p1 + \kappa 1 - p1\, \kappa 1 + 2\, h\, p1\, \kappa 1)}\right\}\right\}$

In[99]:=
```
s2RuleSiblyAlt = Solve[κ2 == p2SelSiblyGiven /. q2 → 1 - p2, s2]
```
Out[99]= $\left\{\left\{s2 \to \frac{-p2 + \kappa 2}{(-1 + p2)\,(-h\, p2 + \kappa 2 - p2\, \kappa 2 + 2\, h\, p2\, \kappa 2)}\right\}\right\}$

Using the symmetrical parameterisation

In[100]:=
```
s1RuleSymmet = Solve[κ1 == p1SelSymmet, s1]
```
Out[100]= $\left\{\left\{s1 \to \frac{p1 - \kappa 1}{(-1 + p1)\,(-h\, p1 + \kappa 1 - p1\, \kappa 1 + 2\, h\, p1\, \kappa 1)}\right\}\right\}$

In[101]:=
```
s2RuleSymmet = Solve[κ2 == p2SelSymmet, s2]
```
Out[101]= $\left\{\left\{s2 \to \frac{p2 - \kappa 2}{p2\,(h + p2 - h\, p2 - 2\, h\, \kappa 2 - p2\, \kappa 2 + 2\, h\, p2\, \kappa 2)}\right\}\right\}$

In[102]:=
```
(s1 /. s1RuleSiblyAlt[[1]]) - (-s1 /. s1RuleSymmet[[1]]) // FullSimplify
```
Out[102]= 0

In[103]:=
```
(s2 /. s2RuleSiblyAlt[[1]]) - (-s2 /. s2RuleSymmet[[1]]) // FullSimplify
```
Out[103]= $-\frac{(p2 - \kappa 2)\,\left(p2^2 + \kappa 2 - 2\, p2\, \kappa 2\right)}{(-1 + p2)\, p2\,(p2 - p2\, \kappa 2 + h\,(-1 + p2) \times (-1 + 2\, \kappa 2))\,(\kappa 2 - p2\, \kappa 2 + h\, p2\,(-1 + 2\, \kappa 2))}$

As expected based on the nature of the two parameterisations, $s_1^{(\mathrm{Sibly})} = -s_1^{(\mathrm{Symmet})}$, but the relationship between $s_2^{(\mathrm{Sibly})}$ and $s_2^{(\mathrm{Symmet})}$ is more complicated.

Validation of Equations 6 and 7 of Sibly et al. (2021) Theor Popul Biol

Using the parameterisation of Sibly et al. (2021)

We follow Sibly et al. (2021) and compute the equilibrium frequencies of alleles given known selection coefficients.

Equating the two expressions for p_1' and solving for p_1 yields Eq. (4a).

```
In[104]:= p1Rule = Solve[(p1Prime /. p1PrimeEq3LRule) == (p1Prime /. p1PrimeEq3RRule), p1]
```

Out[104]= $\left\{\left\{p1 \to \frac{c2\ p2 - c2\ m12\ p2 - p2Prime + m12\ p2Prime + m21\ p2Prime}{c1\ m12}\right\}\right\}$

Substituting from Eq. (1b) for p_2' yields

```
In[105]:= p1SiblyRaw = p1 /. p1Rule[[1]] /. {p2Prime → p2SelSiblyGiven}
```

Out[105]= $\frac{c2\ p2 - c2\ m12\ p2 - \frac{p2^2+p2\ q2\ (1+h\ s2)}{1+2\ h\ p2\ q2\ s2+q2^2\ s2} + \frac{m12\ (p2^2+p2\ q2\ (1+h\ s2))}{1+2\ h\ p2\ q2\ s2+q2^2\ s2} + \frac{m21\ (p2^2+p2\ q2\ (1+h\ s2))}{1+2\ h\ p2\ q2\ s2+q2^2\ s2}}{c1\ m12}$

In[106]:= $p1SiblyGiven = \frac{m12 + m21 - 1}{c1\ m12}\ \frac{p2\ (1 + h\ s2\ q2)}{1 + 2\ h\ s2\ p2\ q2 + s2\ q2^2} + \frac{1 - m12}{c1\ m12}\ c2\ p2$

Out[106]= $\frac{c2\ (1 - m12)\ p2}{c1\ m12} + \frac{(-1 + m12 + m21)\ p2\ (1 + h\ q2\ s2)}{c1\ m12\ (1 + 2\ h\ p2\ q2\ s2 + q2^2\ s2)}$

```
In[107]:= (p1SiblyRaw - p1SiblyGiven) /. q2 → (1 - p2) // FullSimplify
Out[107]= 0
```

```
In[108]:= p2Rule = Solve[(p2Prime /. p2PrimeEq3LRule) == (p2Prime /. p2PrimeEq3RRule), p2]
```

Out[108]= $\left\{\left\{p2 \to \frac{c1\ p1 - c1\ m21\ p1 - p1Prime + m12\ p1Prime + m21\ p1Prime}{c2\ m21}\right\}\right\}$

Substituting from Eq. (1b) for p_2' yields

```
In[109]:= p2SiblyRaw = p2 /. p2Rule[[1]] /. {p1Prime → p1SelSiblyGiven}
```

Out[109]= $\frac{c1\ p1 - c1\ m21\ p1 - \frac{p1^2+p1\ q1\ (1+h\ s1)}{1+2\ h\ p1\ q1\ s1+q1^2\ s1} + \frac{m12\ (p1^2+p1\ q1\ (1+h\ s1))}{1+2\ h\ p1\ q1\ s1+q1^2\ s1} + \frac{m21\ (p1^2+p1\ q1\ (1+h\ s1))}{1+2\ h\ p1\ q1\ s1+q1^2\ s1}}{c2\ m21}$

In[110]:= $p2SiblyGiven = \frac{m12 + m21 - 1}{c2\ m21}\ \frac{p1\ (1 + h\ s1\ q1)}{1 + 2\ h\ s1\ p1\ q1 + s1\ q1^2} + \frac{1 - m21}{c2\ m21}\ c1\ p1$

Out[110]= $\frac{c1\ (1 - m21)\ p1}{c2\ m21} + \frac{(-1 + m12 + m21)\ p1\ (1 + h\ q1\ s1)}{c2\ m21\ (1 + 2\ h\ p1\ q1\ s1 + q1^2\ s1)}$

```
In[111]:= (p2SiblyRaw - p2SiblyGiven) /. q1 → (1 - p1) // FullSimplify
Out[111]= 0
```

The steps above validate Eq. (7) in Sibly et al. (2021).

Using the symmetrical parameterisation

We follow Sibly et al. (2021) and compute the equilibrium frequencies of alleles given known selection coefficients.

Equating the two expressions for p_1' and solving for p_1 yields Eq. (4a).

Substituting from Eq. (1b) for p_2' yields

In[112]:=
```
p1SymmetRaw = p1 /. p1Rule[[1]] /. {p2Prime → p2SelSymmet}
```

Out[112]=
$$\frac{c2\ p2 - c2\ m12\ p2 - \frac{p2\ (1+h\ (-1+p2)\ s2-p2\ s2)}{1+2\ h\ (-1+p2)\ p2\ s2-p2^2\ s2} + \frac{m12\ p2\ (1+h\ (-1+p2)\ s2-p2\ s2)}{1+2\ h\ (-1+p2)\ p2\ s2-p2^2\ s2} + \frac{m21\ p2\ (1+h\ (-1+p2)\ s2-p2\ s2)}{1+2\ h\ (-1+p2)\ p2\ s2-p2^2\ s2}}{c1\ m12}$$

In[113]:=
$$p1Symmet = \frac{m12 + m21 - 1}{c1\ m12}\ \frac{p2 - p2^2\ s2 - h\ p2\ q2\ s2}{1 - 2\ h\ p2\ q2\ s2 - p2^2\ s2} + \frac{1 - m12}{c1\ m12}\ c2\ p2$$

Out[113]=
$$\frac{c2\ (1 - m12)\ p2}{c1\ m12} + \frac{(-1 + m12 + m21)\ (p2 - p2^2\ s2 - h\ p2\ q2\ s2)}{c1\ m12\ (1 - p2^2\ s2 - 2\ h\ p2\ q2\ s2)}$$

In[114]:=
```
(p1SymmetRaw - p1Symmet) /. q2 → (1 - p2) // FullSimplify
```

Out[114]= 0

Substituting from Eq. (1b) for $p_2^{'}$ yields

In[115]:=
```
p2SymmetRaw = p2 /. p2Rule[[1]] /. {p1Prime → p1SelSymmet}
```

Out[115]=
$$\frac{c1\ p1 - c1\ m21\ p1 - \frac{p1+h\ (-1+p1)\ p1\ s1}{1+(-1+p1)\times(1+(-1+2\ h)\ p1)\ s1} + \frac{m12\ (p1+h\ (-1+p1)\ p1\ s1)}{1+(-1+p1)\times(1+(-1+2\ h)\ p1)\ s1} + \frac{m21\ (p1+h\ (-1+p1)\ p1\ s1)}{1+(-1+p1)\times(1+(-1+2\ h)\ p1)\ s1}}{c2\ m21}$$

In[116]:=
$$p2Symmet = \frac{m12 + m21 - 1}{c2\ m21}\ \frac{p1 - h\ p1\ q1\ s1}{1 - 2\ h\ p1\ q1\ s1 - q1^2\ s1} + \frac{1 - m21}{c2\ m21}\ c1\ p1$$

Out[116]=
$$\frac{c1\ (1 - m21)\ p1}{c2\ m21} + \frac{(-1 + m12 + m21)\ (p1 - h\ p1\ q1\ s1)}{c2\ m21\ (1 - 2\ h\ p1\ q1\ s1 - q1^2\ s1)}$$

In[117]:=
```
(p2SymmetRaw - p2Symmet) /. q1 → (1 - p1) // FullSimplify
```

Out[117]= 0

Numerical solution for equilibrium allele frequencies

Remark

There are multiple types of equilibria. Two trivial equilibria correspond to the global loss $(\hat{p}_1 = \hat{p}_2 = 0)$ and fixation $(\hat{p}_1 = \hat{p}_2 = 1)$ of the focal allele A_1. Two further, marginal, equilibria correspond to the case where A_1 is fixed in one and lost in the other deme $(\hat{p}_1 = 1, \hat{p}_2 = 0;\ \hat{p}_1 = 0, \hat{p}_2 = 1)$. Last, there may be an equilibrium at which $A_1$is polymorphic in both demes $(0 < \hat{p}_1, \hat{p}_2 < 1)$.

Using parameterisation by Sibly et al. (2021)

From Eq. (7) of Sibly et al. (2021):

In[274]:=
$$EqPSibly1 := p1 == \frac{m12 + m21 - 1}{c1\ m12}\ \frac{p2\ (1 + h\ s2\ q2)}{1 + 2\ h\ s2\ p2\ q2 + s2\ q2^2} + \frac{1 - m12}{c1\ m12}\ c2\ p2$$

$$EqPSibly2 := p2 == \frac{m12 + m21 - 1}{c2\ m21}\ \frac{p1\ (1 + h\ s1\ q1)}{1 + 2\ h\ s1\ p1\ q1 + s1\ q1^2} + \frac{1 - m21}{c2\ m21}\ c1\ p1$$

Specific test values of the parameters, as per Fig. 2 of Sibly et al. (2021):

In[276]:=
```
mTest1 := 0.05
s1Test1 := -0.01
s1Test1Alt := 0.01
s2Test1 := 0.01
hTest1 := 1.
hTest1Alt := 0.
```

In[282]:=
```
{EqPSibly1, EqPSibly2, 0 ≤ p1 ≤ 1, 0 ≤ p2 ≤ 1} /. qRule /. cRule /. symmetMigRule
```

Out[282]=
$$\left\{p1 == \frac{(1-m)\,p2}{m} + \frac{(-1+2\,m)\,p2\,(1+h\,(1-p2)\,s2)}{m\left(1+(1-p2)^2\,s2+2\,h\,(1-p2)\,p2\,s2\right)},\right.$$
$$\left.p2 == \frac{(1-m)\,p1}{m} + \frac{(-1+2\,m)\,p1\,(1+h\,(1-p1)\,s1)}{m\left(1+(1-p1)^2\,s1+2\,h\,(1-p1)\,p1\,s1\right)},\ 0 \le p1 \le 1,\ 0 \le p2 \le 1\right\}$$

In[283]:=
```
NSolveSibly[m_, s1_, s2_, h_] := NSolve[
  {p1 == (1 - m) p2 / m + (-1 + 2 m) p2 (1 + h (1 - p2) s2) / (m (1 + (1 - p2)^2 s2 + 2 h (1 - p2) p2 s2)),
   p2 == (1 - m) p1 / m + (-1 + 2 m) p1 (1 + h (1 - p1) s1) / (m (1 + (1 - p1)^2 s1 + 2 h (1 - p1) p1 s1)),
   0 ≤ p1 ≤ 1, 0 ≤ p2 ≤ 1}, {p1, p2},
  Reals
 ]
```

In[284]:=
```
NSolveSibly[mTest1, s1Test1, s2Test1, hTest1]
```

Out[284]=
$$\{\{p1 \to 0.707456,\ p2 \to 0.680969\},\ \{p1 \to 7.37089\times10^{-10},\ p2 \to 7.37089\times10^{-10}\},\ \{p1 \to 0.,\ p2 \to 0.\}\}$$

In[285]:=
```
Manipulate[
 NSolveSibly[m, s1, s2, h] // Chop,
 {{m, mTest1}, 0, 1}, {{s1, s1Test1}, -1., 100},
 {{s2, s2Test1}, -1., 100}, {{h, hTest1}, 0, 1}
]
```

Out[285]=



$$\{\{p1 \to 0.707456,\ p2 \to 0.680969\},\ \{p1 \to 7.37089\times10^{-10},\ p2 \to 7.37089\times10^{-10}\},\ \{p1 \to 0,\ p2 \to 0\}\}$$

With the parameter values used by Sibly et al. (2021) in their Fig. 2 ($h = 1$, $m = 0.05$, $s_1 = -0.01$, $s_2 = 0.01$), the internal equilibrium is given by the first pair of coordinates in the output above. We store these values for reference in a plot below:

In[286]:=
```
pEqValSiblyFig2 = NSolveSibly[mTest1, s1Test1, s2Test1, hTest1][[1]]
```
Out[286]=
```
{p1 → 0.707456, p2 → 0.680969}
```

Using symmetrical parameterisation

In[287]:=

$$\text{EqPSymmet1} := \text{p1} == \frac{\text{m12}+\text{m21}-1}{\text{c1 m12}}\,\frac{\text{p2}-\text{p2}^2\,\text{s2}-\text{h p2 q2 s2}}{1-2\,\text{h p2 q2 s2}-\text{p2}^2\,\text{s2}}+\frac{1-\text{m12}}{\text{c1 m12}}\,\text{c2 p2}$$

$$\text{EqPSymmet2} := \text{p2} == \frac{\text{m12}+\text{m21}-1}{\text{c2 m21}}\,\frac{\text{p1}-\text{h p1 q1 s1}}{1-2\,\text{h p1 q1 s1}-\text{q1}^2\,\text{s1}}+\frac{1-\text{m21}}{\text{c2 m21}}\,\text{c1 p1}$$

In[289]:=
```
{EqPSymmet1, EqPSymmet2, 0 ≤ p1 ≤ 1, 0 ≤ p2 ≤ 1} /. qRule /. cRule /. symmetMigRule
```
Out[289]=

$$\left\{\text{p1} == \frac{(1-\text{m})\,\text{p2}}{\text{m}}+\frac{(-1+2\,\text{m})\left(\text{p2}-\text{h}\,(1-\text{p2})\,\text{p2 s2}-\text{p2}^2\,\text{s2}\right)}{\text{m}\left(1-2\,\text{h}\,(1-\text{p2})\,\text{p2 s2}-\text{p2}^2\,\text{s2}\right)},\right.$$
$$\left.\text{p2} == \frac{(1-\text{m})\,\text{p1}}{\text{m}}+\frac{(-1+2\,\text{m})\,(\text{p1}-\text{h}\,(1-\text{p1})\,\text{p1 s1})}{\text{m}\left(1-(1-\text{p1})^2\,\text{s1}-2\,\text{h}\,(1-\text{p1})\,\text{p1 s1}\right)},\ 0\le\text{p1}\le 1,\ 0\le\text{p2}\le 1\right\}$$

In[290]:=

$$\text{NSolveSymmet[m\_, s1\_, s2\_, h\_]} := \text{NSolve}\Big[$$
$$\left\{\text{p1} == \frac{(1-\text{m})\,\text{p2}}{\text{m}}+\frac{(-1+2\,\text{m})\left(\text{p2}-\text{h}\,(1-\text{p2})\,\text{p2 s2}-\text{p2}^2\,\text{s2}\right)}{\text{m}\left(1-2\,\text{h}\,(1-\text{p2})\,\text{p2 s2}-\text{p2}^2\,\text{s2}\right)},\right.$$
$$\text{p2} == \frac{(1-\text{m})\,\text{p1}}{\text{m}}+\frac{(-1+2\,\text{m})\,(\text{p1}-\text{h}\,(1-\text{p1})\,\text{p1 s1})}{\text{m}\left(1-(1-\text{p1})^2\,\text{s1}-2\,\text{h}\,(1-\text{p1})\,\text{p1 s1}\right)},$$
$$\left.0\le\text{p1}\le 1,\ 0\le\text{p2}\le 1\right\},\ \{\text{p1},\ \text{p2}\},$$
$$\text{Reals}$$
$$\Big]$$

In[291]:=
```
NSolveSymmet[mTest1, s1Test1Alt, s2Test1, hTest1Alt]
```
Out[291]=
```
{{p1 → 1., p2 → 1.}, {p1 → 0.511023, p2 → 0.488977}, {p1 → 0., p2 → 0.}}
```

In[292]:=
```
Manipulate[
 NSolveSymmet[m, s1, s2, h] // Chop,
 {{m, mTest1}, 0, 1}, {{s1, s1Test1Alt}, 0., 1.},
 {{s2, s2Test1}, 0., 1.}, {{h, hTest1Alt}, 0, 1}
]
```

Out[292]=




```
{{p1 → 1., p2 → 1.}, {p1 → 0.511023, p2 → 0.488977}, {p1 → 0, p2 → 0}}
```

In[293]:=
```
pEqValSymmet = NSolveSymmet[mTest1, s1Test1Alt, s2Test1, hTest1Alt][[2]]
```

Out[293]=
```
{p1 → 0.511023, p2 → 0.488977}
```

Recursion equations for migration and selection

Parameterisation by Sibly et al. (2021)

In[294]:=
```
p1PrimePrime
```

Out[294]=

$$\frac{(1 - m12)\ p1Prime + m21\ p2Prime}{c1}$$

In[295]:=
```
p2PrimePrime
```

Out[295]=

$$\frac{m12\ p1Prime + (1 - m21)\ p2Prime}{c2}$$

In[333]:=
```
p1SelSibly
```

Out[333]=

$$\frac{p1\ (-1 + h\ (-1 + p1)\ s1)}{-1 + (-1 + p1) \times (1 + (-1 + 2\ h)\ p1)\ s1}$$

In[297]:=
```
p2SelSibly
```

Out[297]=

$$\frac{p2\ (-1 + h\ (-1 + p2)\ s2)}{-1 + (-1 + p2) \times (1 + (-1 + 2\ h)\ p2)\ s2}$$

In[298]:=
```
p1PrimePrime /. {p1Prime → p1SelSibly, p2Prime → p2SelSibly} /. cRule /. qRule //
  Simplify
p2PrimePrime /. {p1Prime → p1SelSibly, p2Prime → p2SelSibly} /. cRule /. qRule //
  Simplify
```

Out[298]=

$$\frac{\frac{(1-m12)\ p1\ (-1+h\ (-1+p1)\ s1)}{-1+(-1+p1)\times(1+(-1+2\ h)\ p1)\ s1}+\frac{m21\ p2\ (-1+h\ (-1+p2)\ s2)}{-1+(-1+p2)\times(1+(-1+2\ h)\ p2)\ s2}}{1-m12+m21}$$

Out[299]=

$$\frac{\frac{m12\ p1\ (-1+h\ (-1+p1)\ s1)}{-1+(-1+p1)\times(1+(-1+2\ h)\ p1)\ s1}+\frac{(1-m21)\ p2\ (-1+h\ (-1+p2)\ s2)}{-1+(-1+p2)\times(1+(-1+2\ h)\ p2)\ s2}}{1+m12-m21}$$

Initial allele frequencies as per Sibly et al. (“initial frequency of Q of 0.01”, which translates to $p_1 = p_2 = 0.99$):

In[300]:=
```
p1Init1 := 0.99
p2Init1 := 0.99
```

In[302]:=
```
Remove[p1RecSibly, p2RecSibly]
p1RecSibly[m12_, m21_, s1_, s2_, h_, 0] := p1Init1
p2RecSibly[m12_, m21_, s1_, s2_, h_, 0] := p2Init1

p1RecSibly[m12_, m21_, s1_, s2_, h_, t_] := p1RecSibly[m12, m21, s1, s2, h, t] =
   1/(1 - m12 + m21) (((1 - m12) p1RecSibly[m12, m21, s1, s2, h, t - 1]
         (-1 + h (-1 + p1RecSibly[m12, m21, s1, s2, h, t - 1]) s1)) /
       (-1 + (-1 + p1RecSibly[m12, m21, s1, s2, h, t - 1]) ×
          (1 + (-1 + 2 h) p1RecSibly[m12, m21, s1, s2, h, t - 1]) s1) +
      (m21 p2RecSibly[m12, m21, s1, s2, h, t - 1]
         (-1 + h (-1 + p2RecSibly[m12, m21, s1, s2, h, t - 1]) s2)) /
       (-1 + (-1 + p2RecSibly[m12, m21, s1, s2, h, t - 1]) ×
          (1 + (-1 + 2 h) p2RecSibly[m12, m21, s1, s2, h, t - 1]) s2))
p2RecSibly[m12_, m21_, s1_, s2_, h_, t_] := p2RecSibly[m12, m21, s1, s2, h, t] =
   1/(1 + m12 - m21) ((m12 p1RecSibly[m12, m21, s1, s2, h, t - 1]
         (-1 + h (-1 + p1RecSibly[m12, m21, s1, s2, h, t - 1]) s1)) /
       (-1 + (-1 + p1RecSibly[m12, m21, s1, s2, h, t - 1]) ×
          (1 + (-1 + 2 h) p1RecSibly[m12, m21, s1, s2, h, t - 1]) s1) +
      ((1 - m21) p2RecSibly[m12, m21, s1, s2, h, t - 1]
         (-1 + h (-1 + p2RecSibly[m12, m21, s1, s2, h, t - 1]) s2)) /
       (-1 + (-1 + p2RecSibly[m12, m21, s1, s2, h, t - 1]) ×
          (1 + (-1 + 2 h) p2RecSibly[m12, m21, s1, s2, h, t - 1]) s2))
```

In[307]:=
```
DiscretePlot[
 {
  p1RecSibly[mTest1, mTest1, s1Test1, s2Test1, hTest1, t],
  p2RecSibly[mTest1, mTest1, s1Test1, s2Test1, hTest1, t]
 },
 {t, 0, 10 000},
 GridLines → {None, {p1, p2} /. pEqValSiblyFig2},
 PlotRange → {Full, Full},
 Filling → None,
 Joined → True,
 Frame → True,
 FrameLabel → {"Generation t", "Frequency of allele A1"},
 PlotLegends → {"Deme 1", "Deme 2"}
]
```

Out[307]=



The plot above should correspond to Fig. 3 in Sibly et al. (2021), but it does not.

The same plot as above, but extending the x axis to 50,000 generations to see if the equilibrium allele frequencies as determined by numerical solution are reached:

In[308]:=
```
DiscretePlot[
 {
  p1RecSibly[mTest1, mTest1, s1Test1, s2Test1, hTest1, t],
  p2RecSibly[mTest1, mTest1, s1Test1, s2Test1, hTest1, t]
 },
 {t, 0, 50 000},
 GridLines → {None, {p1, p2} /. pEqValSiblyFig2},
 PlotRange → {Full, Full},
 Filling → None,
 Joined → True,
 Frame → True,
 FrameLabel → {"Generation t", "Frequency of allele A1"},
 PlotLegends → {"Deme 1", "Deme 2"}
]
```

Out[308]=



Figure 2 in Sibly et al. (2021) seems to illustrate the dynamics for another parameter combination than the one reported, i.e. not for $m = 0.05$, $s_1 = -0.01$, $s_2 = 0.01$, $h = 1.0$. Alternatively, the might be an error in the implementation of the simulations by Sibly et al. (2020).

Symmetrical parameterisation

In[309]:= `p1PrimePrime`

Out[309]= $$\frac{(1 - m12)\ p1Prime + m21\ p2Prime}{c1}$$

In[310]:= `p2PrimePrime`

Out[310]= $$\frac{m12\ p1Prime + (1 - m21)\ p2Prime}{c2}$$

In[311]:= `p1SelSymmet`

Out[311]= $$\frac{p1 + h\ (-1 + p1)\ p1\ s1}{1 + (-1 + p1) \times (1 + (-1 + 2\ h)\ p1)\ s1}$$

In[312]:=
```
p2SelSymmet
```

Out[312]=
$$\frac{p2\ (1 + h\ (-1 + p2)\ s2 - p2\ s2)}{1 + 2\ h\ (-1 + p2)\ p2\ s2 - p2^2\ s2}$$

In[313]:=
```
(1 - (p2SelSymmet /. {p2 → 1 - p2} /. {p2 → p1, s2 → s1})) - p1SelSymmet // FullSimplify
```

Out[313]= 0

In[314]:=
```
qRule
```

Out[314]= {q1 → 1 - p1, q2 → 1 - p2}

In[315]:=
```
p1SelMigSymmet =
 p1PrimePrime /. {p1Prime → p1SelSymmet, p2Prime → p2SelSymmet} /. cRule //
  Simplify
p2SelMigSymmet = p2PrimePrime /. {p1Prime → p1SelSymmet, p2Prime → p2SelSymmet} /.
   cRule // Simplify
```

Out[315]=
$$\frac{\frac{(1-m12)\ (p1+h\ (-1+p1)\ p1\ s1)}{1+(-1+p1)\times(1+(-1+2\ h)\ p1)\ s1}+\frac{m21\ p2\ (1+h\ (-1+p2)\ s2-p2\ s2)}{1+2\ h\ (-1+p2)\ p2\ s2-p2^2\ s2}}{1 - m12 + m21}$$

Out[316]=
$$\frac{\frac{m12\ (p1+h\ (-1+p1)\ p1\ s1)}{1+(-1+p1)\times(1+(-1+2\ h)\ p1)\ s1}+\frac{(1-m21)\ p2\ (1+h\ (-1+p2)\ s2-p2\ s2)}{1+2\ h\ (-1+p2)\ p2\ s2-p2^2\ s2}}{1 + m12 - m21}$$

Initial allele frequencies as per Sibly et al. (“initial frequency of Q of 0.01”, which translates to $p_1 = p_2 = 0.99$), as above.

In[317]:=
```
p1Init1 := 0.99
p2Init1 := 0.99
```

In[319]:=

```
Remove[p1RecSymmet, p2RecSymmet]
p1RecSymmet[m12_, m21_, s1_, s2_, h_, 0] := p1Init1
p2RecSymmet[m12_, m21_, s1_, s2_, h_, 0] := p2Init1

p1RecSymmet[m12_, m21_, s1_, s2_, h_, t_] :=
 p1RecSymmet[m12, m21, s1, s2, h, t] = 1/(1 - m12 + m21)
   (((1 - m12) (p1RecSymmet[m12, m21, s1, s2, h, t - 1] + h (-1 + p1RecSymmet[m12,
              m21, s1, s2, h, t - 1]) p1RecSymmet[m12, m21, s1, s2, h, t - 1] s1)) /
       (1 + (-1 + p1RecSymmet[m12, m21, s1, s2, h, t - 1]) ×
          (1 + (-1 + 2 h) p1RecSymmet[m12, m21, s1, s2, h, t - 1]) s1) +
      (1 / (1 + 2 h (-1 + p2RecSymmet[m12, m21, s1, s2, h, t - 1]) p2RecSymmet[m12,
             m21, s1, s2, h, t - 1] s2 - p2RecSymmet[m12, m21, s1, s2, h, t - 1]^2 s2))
       m21 p2RecSymmet[m12, m21, s1, s2, h, t - 1]
       (1 + h (-1 + p2RecSymmet[m12, m21, s1, s2, h, t - 1]) s2 -
         p2RecSymmet[m12, m21, s1, s2, h, t - 1] s2))
p2RecSymmet[m12_, m21_, s1_, s2_, h_, t_] :=
 p2RecSymmet[m12, m21, s1, s2, h, t] = 1/(1 + m12 - m21)
   ((m12 (p1RecSymmet[m12, m21, s1, s2, h, t - 1] + h (-1 + p1RecSymmet[m12, m21, s1,
              s2, h, t - 1]) p1RecSymmet[m12, m21, s1, s2, h, t - 1] s1)) /
       (1 + (-1 + p1RecSymmet[m12, m21, s1, s2, h, t - 1]) ×
          (1 + (-1 + 2 h) p1RecSymmet[m12, m21, s1, s2, h, t - 1]) s1) +
      (1 / (1 + 2 h (-1 + p2RecSymmet[m12, m21, s1, s2, h, t - 1])
            p2RecSymmet[m12, m21, s1, s2, h, t - 1] s2 -
          p2RecSymmet[m12, m21, s1, s2, h, t - 1]^2 s2)) ×
       (1 - m21) p2RecSymmet[m12, m21, s1, s2, h, t - 1]
       (1 + h (-1 + p2RecSymmet[m12, m21, s1, s2, h, t - 1]) s2 -
         p2RecSymmet[m12, m21, s1, s2, h, t - 1] s2))
```

In[324]:=

```
mTest1 := 0.05
s1Test1 := -0.01
s1Test1Alt := 0.01
s2Test1 := 0.01
hTest1 := 1.
hTest1Alt := 0.
```

In[330]:=
```
DiscretePlot[
 {
  p1RecSymmet[mTest1, mTest1, s1Test1Alt, s2Test1, hTest1Alt, t],
  p2RecSymmet[mTest1, mTest1, s1Test1Alt, s2Test1, hTest1Alt, t]
 },
 {t, 0, 10 000},
 GridLines → {None, {p1, p2} /. pEqValSymmet},
 PlotRange → {Full, Full},
 Filling → None,
 Joined → True,
 Frame → True,
 FrameLabel → {"Generation t", "Frequency of allele A1"},
 PlotLegends → {"Deme 1", "Deme 2"}
]
```

Out[330]=



The same plot as above, but extending the x axis to 50,000 generations to see if the equilibrium allele frequencies as determined by numerical solution are reached:

In[331]:=
```
DiscretePlot[
 {
  p1RecSymmet[mTest1, mTest1, s1Test1Alt, s2Test1, hTest1Alt, t],
  p2RecSymmet[mTest1, mTest1, s1Test1Alt, s2Test1, hTest1Alt, t]
 },
 {t, 0, 50 000},
 GridLines → {None, {p1, p2} /. pEqValSymmet},
 PlotRange → {Full, Full},
 Filling → None,
 Joined → True,
 Frame → True,
 FrameLabel → {"Generation t", "Frequency of allele A1"},
 PlotLegends → {"Deme 1", "Deme 2"}
]
```

Out[331]=



Not used

Formulae for F_{ST}

In[•]:=
```
FstSibly = (pT (1 - pT) - 1/2 p1 (1 - p1) - 1/2 p2 (1 - p2)) / (pT (1 - pT)) /.
    {pT → (p1SiblyGiven + p2SiblyGiven), p1 → p1SiblyGiven,
     p2 → p2SiblyGiven} /. {q1 → 1 - p1, q2 → 1 - p2} // Simplify
```

Out[•]= $Aborted

In[•]:=
```
FstSymmet =
  (pT (1 - pT) - 1/2 p1 (1 - p1) - 1/2 p2 (1 - p2)) / (pT (1 - pT)) /. {pT → (p1Symmet + p2Symmet),
     p1 → p1Symmet, p2 → p2Symmet} /. {q1 → 1 - p1, q2 → 1 - p2} // Simplify
```

Out[•]=

$$
\begin{aligned}
&\left(-\frac{1}{2\,c2\,m21}\left(-c1\,(-1+m21)\,p1+\frac{(-1+m12+m21)\,(p1+h\,(-1+p1)\,p1\,s1)}{1+(-1+p1)\times(1+(-1+2\,h)\,p1)\,s1}\right)\right.\\
&\quad\left(1+\frac{c1\,(-1+m21)\,p1}{c2\,m21}-\frac{(-1+m12+m21)\,(p1+h\,(-1+p1)\,p1\,s1)}{c2\,m21\,(1+(-1+p1)\times(1+(-1+2\,h)\,p1)\,s1)}\right)+\\
&\quad\left(1+\frac{c1\,(-1+m21)\,p1}{c2\,m21}+\frac{c2\,(-1+m12)\,p2}{c1\,m12}-\frac{(-1+m12+m21)\,(p1+h\,(-1+p1)\,p1\,s1)}{c2\,m21\,(1+(-1+p1)\times(1+(-1+2\,h)\,p1)\,s1)}-\right.\\
&\quad\quad\left.\frac{(-1+m12+m21)\,p2\,(1+h\,(-1+p2)\,s2-p2\,s2)}{c1\,m12\,\left(1+2\,h\,(-1+p2)\,p2\,s2-p2^{2}\,s2\right)}\right)\\
&\quad\left(-\frac{c1\,(-1+m21)\,p1}{c2\,m21}-\frac{c2\,(-1+m12)\,p2}{c1\,m12}+\frac{(-1+m12+m21)\,(p1+h\,(-1+p1)\,p1\,s1)}{c2\,m21\,(1+(-1+p1)\times(1+(-1+2\,h)\,p1)\,s1)}+\right.\\
&\quad\quad\left.\frac{(-1+m12+m21)\,p2\,(1+h\,(-1+p2)\,s2-p2\,s2)}{c1\,m12\,\left(1+2\,h\,(-1+p2)\,p2\,s2-p2^{2}\,s2\right)}\right)+\\
&\quad\Big(p2\,\Big(-((-1+m12+m21)\times(1+h\,(-1+p2)\,s2-p2\,s2))+c2\,(-1+m12)\times\\
&\quad\quad\left(1+2\,h\,(-1+p2)\,p2\,s2-p2^{2}\,s2\right)\Big)\,\Big(c1\,m12\,\left(1+2\,h\,(-1+p2)\,p2\,s2-p2^{2}\,s2\right)+\\
&\quad\quad p2\,\Big(-((-1+m12+m21)\times(1+h\,(-1+p2)\,s2-p2\,s2))+\\
&\quad\quad\quad c2\,(-1+m12)\times\left(1+2\,h\,(-1+p2)\,p2\,s2-p2^{2}\,s2\right)\Big)\Big)\Big)\Big/\\
&\quad\left.\left(2\,c1^{2}\,m12^{2}\,\left(1+2\,h\,(-1+p2)\,p2\,s2-p2^{2}\,s2\right)^{2}\right)\right)\Big/\\
&\left(\left(1+\frac{c1\,(-1+m21)\,p1}{c2\,m21}+\frac{c2\,(-1+m12)\,p2}{c1\,m12}-\frac{(-1+m12+m21)\,(p1+h\,(-1+p1)\,p1\,s1)}{c2\,m21\,(1+(-1+p1)\times(1+(-1+2\,h)\,p1)\,s1)}-\right.\right.\\
&\quad\left.\frac{(-1+m12+m21)\,p2\,(1+h\,(-1+p2)\,s2-p2\,s2)}{c1\,m12\,\left(1+2\,h\,(-1+p2)\,p2\,s2-p2^{2}\,s2\right)}\right)\\
&\quad\left(-\frac{c1\,(-1+m21)\,p1}{c2\,m21}-\frac{c2\,(-1+m12)\,p2}{c1\,m12}+\frac{(-1+m12+m21)\,(p1+h\,(-1+p1)\,p1\,s1)}{c2\,m21\,(1+(-1+p1)\times(1+(-1+2\,h)\,p1)\,s1)}+\right.\\
&\quad\quad\left.\left.\frac{(-1+m12+m21)\,p2\,(1+h\,(-1+p2)\,s2-p2\,s2)}{c1\,m12\,\left(1+2\,h\,(-1+p2)\,p2\,s2-p2^{2}\,s2\right)}\right)\right)
\end{aligned}
$$